Tan's Calculus section 14.6 problem #25: Find the volume of the region bounded by $x = 0$, $z = 0$, $x + z = 4$, and $x = 4 - y^2$.

We know that $x = 0$ and $z = 0$ are coordinate planes. $x + z = 4$ is a plane and $x = 4 - y^2$ is a parabolic cylinder. Let's plot the part of the plane which lies above the region bounded by $x = 4 - y^2$ and $x = 0$.

```
> with(plots):
> plot3d(4-x,x=0..4-y^2,y=-2..2,axes=boxed);
```



Let plot all four surfaces together...

```
> S1 := plot3d(4-x,x=0..4,y=-2..2):
  S2 := plot3d(0,x=0..4,y=-2..2):
  S3 := plot3d([0,y,z],y=-2..2,z=0..4):
  S4 := plot3d([4-y^2,y,z],y=-2..2,z=0..4):
  display({S1,S2,S3,S4});
```



Now cut down the surfaces...

Note that the upper bound for z on the final surface is the intersection of $z = 4 - x$ and $x = 4 - y^2$.

```
> S1 := plot3d(4-x,x=0..4-y^2,y=-2..2):
  S2 := plot3d(0,x=0..4-y^2,y=-2..2):
```

```
S3 := plot3d([0,y,z],y=-2..2,z=0..4):
S4 := plot3d([4-y^2,y,z],y=-2..2,z=0..4-(4-y^2)):
display({S1,S2,S3,S4});
```



The volume is the triple integral of 1.

```
> Int(Int(Int(1,z=0..4-x),x=0..4-y^2),y=-2..2) = int(int(int(1,z=0..4
-x),x=0..4-y^2),y=-2..2);
```

$$\int_{-2}^{2}\int_{0}^{-y^2+4}\int_{0}^{4-x} 1 \, dz \, dx \, dy = \frac{128}{5} \quad \textbf{(1)}$$

An example using rectangular, cylindrical, and spherical coordinates.

```
> with(plots):
> top := plot3d([5*cos(theta)*sin(phi),5*sin(theta)*sin(phi),5*cos
(phi)],phi=0..arctan(3/4),theta=0..2*Pi):
bottom := plot3d([r*cos(theta),r*sin(theta),4],r=0..3,theta=0..2*
Pi):

display({top,bottom},scaling=constrained,axes=boxed);
```



We wish to compute $\iiint_E \sin(x^2+y^2+z^2)\,dV$ where E is the region inside $x^2+y^2+z^2=25$ and above $z=4$.

Note: "int" is Maple's integrate command. "Int" is Maple's "inert" integrate command. The inert command represents an integral, but doesn't actually go through with the computation. To actually evaluate inert commands right-click on the blue output and choose "Evaluate (from inert)".

To compute iterated integrals, we simply stick int commands inside of other int commands.

Rectangular:

```
> Int(Int(Int(sin(x^2+y^2+z^2),z=4..sqrt(25-x^2-y^2)),y=-sqrt(9-x^2).
  .sqrt(9-x^2)),x=-3..3);
```

$$\int_{-3}^{3}\int_{-\sqrt{-x^2+9}}^{\sqrt{-x^2+9}}\int_{4}^{\sqrt{-x^2-y^2+25}} \sin(x^2+y^2+z^2)\,dz\,dy\,dx \qquad (2)$$

```
> # I gave this one a minute or two, but Maple never got back with an
  answer. :(
  # int(int(int(sin(x^2+y^2+z^2),z=4..sqrt(25-x^2-y^2)),y=-sqrt(9-
  x^2)..sqrt(9-x^2)),x=-3..3);
```

Cylindrical:

```
> Int(Int(Int(sin(r^2+z^2)*r,z=4..sqrt(25-r^2)),r=0..3),theta=0..2*
  Pi);
```

$$\int_{0}^{2\pi}\int_{0}^{3}\int_{4}^{\sqrt{-r^2+25}} \sin(r^2+z^2)\,r\,dz\,dr\,d\theta \qquad (3)$$

```
> int(int(int(sin(r^2+z^2)*r,z=4..sqrt(25-r^2)),r=0..3),theta=0..2*
  Pi);
```

(4)

$$-\frac{1}{2}\sqrt{2}\,\pi^{3/2}\,\mathrm{FresnelC}\left(\frac{4\sqrt{2}}{\sqrt{\pi}}\right)+\frac{1}{2}\sqrt{2}\,\pi^{3/2}\,\mathrm{FresnelC}\left(\frac{5\sqrt{2}}{\sqrt{\pi}}\right)-\cos(25)\,\pi \qquad \mathbf{(4)}$$

Spherical:

```
> Int(Int(Int(sin(rho^2)*rho^2*sin(phi),rho=4*sec(phi)..5),phi=0..
  arctan(3/4)),theta=0..2*Pi);
```

$$\int_0^{2\pi}\int_0^{\arctan\left(\frac{3}{4}\right)}\int_{4\sec(\phi)}^{5}\sin\left(\rho^2\right)\rho^2\sin(\phi)\,d\rho\,d\phi\,d\theta \qquad \mathbf{(5)}$$

```
> int(int(int(sin(rho^2)*rho^2*sin(phi),rho=4*sec(phi)..5),phi=0..
  arctan(3/4)),theta=0..2*Pi);
```

$$-\frac{1}{2}\sqrt{2}\,\pi^{3/2}\,\mathrm{FresnelC}\left(\frac{4\sqrt{2}}{\sqrt{\pi}}\right)+\frac{1}{2}\sqrt{2}\,\pi^{3/2}\,\mathrm{FresnelC}\left(\frac{5\sqrt{2}}{\sqrt{\pi}}\right)-\cos(25)\,\pi \qquad \mathbf{(6)}$$

The answer approximated...

```
> evalf(int(int(int(sin(rho^2)*rho^2*sin(phi),rho=4*sec(phi)..5),phi=
  0..arctan(3/4)),theta=0..2*Pi));
```

$$-3.060528180 \qquad \mathbf{(7)}$$

The centroid of our region...

```
> m := int(int(int(1*rho^2*sin(phi),rho=4*sec(phi)..5),phi=0..arctan
  (3/4)),theta=0..2*Pi);
```

$$m := \frac{14}{3}\pi \qquad \mathbf{(8)}$$

```
> Myz := int(int(int(rho*cos(theta)*sin(phi)*rho^2*sin(phi),rho=4*sec
  (phi)..5),phi=0..arctan(3/4)),theta=0..2*Pi);
```

$$Myz := 0 \qquad \mathbf{(9)}$$

```
> Mxz := int(int(int(rho*sin(theta)*sin(phi)*rho^2*sin(phi),rho=4*sec
  (phi)..5),phi=0..arctan(3/4)),theta=0..2*Pi);
```

$$Mxz := 0 \qquad \mathbf{(10)}$$

```
> Mxy := int(int(int(rho*cos(phi)*rho^2*sin(phi),rho=4*sec(phi)..5),
  phi=0..arctan(3/4)),theta=0..2*Pi);
```

$$Mxy := \frac{81}{4}\pi \qquad \mathbf{(11)}$$

```
> Centroid := 1/m*<Myz,Mxz,Mxy> = evalf(1/m*<Myz,Mxz,Mxy>);
```

$$Centroid := \begin{bmatrix} 0 \\ 0 \\ \frac{243}{56} \end{bmatrix} = \begin{bmatrix} 0. \\ 0. \\ 4.339285714 \end{bmatrix} \qquad \mathbf{(12)}$$